

Estruturas de Repetição

Notas de Aula

Prof. Francisco Rapchan
www.geocities.com/chicorapchan
rapchan@terra.com.br

Exemplo 1. Faça um algoritmo que mostre todos os números de 1 até 5.

Uma solução MUITO simples seria:

```
Algoritmo "Números"
Início
    Escreva (1,2,3,4,5)
Fimalgoritmo
```

Uma solução um pouco mais elaborada poderia ser:

```
Algoritmo "Números"
    Var numero : inteiro
Início
    numero <- 1;

    Escreva (numero)
    numero <- numero + 1
    Escreva (numero)
    numero <- numero + 1
    Escreva (numero)
    numero <- numero + 1
    Escreva (numero)
    numero <- numero + 1
    Escreva (numero)
Fimalgoritmo
```

Observe que nesta solução o valor de numero vai sendo **incrementado** (aumentado) ao longo do algoritmo.

Uma solução usando a estrutura de repetição **enquanto** seria:

```
Algoritmo "Números"
    Var numero : inteiro
Início

    numero <- 1;

    Enquanto numero <= 5 faca
        Escreva (numero)
        numero <- numero + 1
    fimenquanto

Fimalgoritmo
```

Exemplo 2. Faça um algoritmo que mostre todos os números pares menores que 15.

```
Algoritmo "O que faz?"
  Var numero : inteiro
Inicio

  numero <- 0;

  Enquanto numero < 15 faca
    Escreva (numero)
    numero <- numero + 2
  fimenquanto

Fimalgoritmo
```

Exemplo 3. Faça um algoritmo que leia dois números e mostre todos os números entre eles (do menor até o maior).

```
algoritmo "Leia os inteiros"

var
  x,y: inteiro

inicio
  Leia (x,y)
  Enquanto x <= y faca
    escreva (x)
    x <- x + 1
  Fimenquanto
fimalgoritmo
```

Exemplo 4. Mostre os pares entre 1 e 10 de forma decrescente

```
Algoritmo "Leia os inteiros"
Var
  numero : inteiro

inicio
  numero <- 10
  enquanto numero >= 2 faca
    escreva (numero )
    numero <- numero - 2
  fimenquanto
fimalgoritmo
```

Exemplo 5. Mostre os pares entre 1 e 10 de forma crescente e decrescente

```
Algoritmo "Mostre pares crescente e decrescente"
Var
  numero : inteiro
Inicio
  numero <- 2
  Escreva ("Números em ordem crescente")
  Enquanto numero <= 10 Faca
    Escreva (numero)
    numero <- numero + 2
  Fimenquanto

  numero<- 10
  Escreva ("Números em ordem crescente")
  Enquanto numero >= 2 Faca
    Escreva (numero)
    numero <- numero - 2
  Fimenquanto
Fimalgoritmo
```

Exemplo 6. Explique o que faz o algoritmo abaixo.

```
Algoritmo "O que faz?"
  Var n, s: Inteiro
Inicio
  n <- 1;
  s <- 0;
  Enquanto n <= 5 faça
    s <- s + n
    n <- n + 1
  Fimenquanto
  Escreva (s)
Fimalgoritmo
```

Para facilitar, faça uma tabela de verificação para as variáveis.

n	s

Exemplo 7. Leia 5 números e mostre a soma e a média deles.

```
algoritmo "Leia 5 números e mostre a soma e a média deles"
var
  contador : inteiro
  numero, soma, media : real
inicio
  contador <- 1
  soma <- 0
  media <- 0
  enquanto contador <= 5 faça
    leia (numero)
    contador <- contador + 1
    soma <- soma + numero
  fimenquanto
  escreva (soma)
  media <- soma/5
  escreva (media)
fimalgoritmo
```

Exemplo 8. Leia n números (ou seja, leia um número determinado dinamicamente de números) e mostre a soma e a média deles.

```
algoritmo "Leia 5 números e mostre a soma e a média deles"
var
  contador, quantidade : inteiro
  numero, soma, media : real
inicio
  Leia (quantidade)
  contador <- 1
  soma <- 0
  media <- 0
  enquanto contador <= quantidade faça
    leia (numero)
    contador <- contador + 1
    soma <- soma + numero
  fimenquanto
  escreva (soma)
  media <- soma/quantidade
  escreva (media)
fimalgoritmo
```

Exemplo 9. Faça um algoritmo que mostre a soma dos 10 primeiros números inteiros.

```
ALGORITMO "Soma dos dez primeiros números"
  var soma, numero: numerico

INICIO
  numero <- 1
  soma <- 0

  Enquanto (numero <= 10) faça
    soma <- soma + numero
    numero <- numero + 1
  Fimenquanto

  escreva (soma)
FIMALGORITMO
```

A idéia básica deste algoritmo é somar os números. Observe que há uma repetição nesta tarefa: some o primeiro número, depois some o segundo e assim por diante até o décimo número. Podemos pensar assim: enquanto o valor da variável numero for menor ou igual a 10, inclua em soma o valor do numero e incremente (some 1) esta variável.

Observe que o algoritmo fica fazendo a soma e incrementando a variável numero enquanto não for 10. Chamamos esta estrutura de repetição de “**ENQUANTO-FAÇA**”. Este mesmo algoritmo pode ser feito usando uma estrutura de repetição chamada “**PARA-FAÇA**”:

```
ALGORITMO "Soma dos dez primeiros números"
  var soma, numero: numerico

INICIO
  soma <- 0

  Para numero de 1 ate 10 Faca
    soma <- soma + numero
  Fimpara

  Eescreva (soma)

FIMALGORITMO
```

Exemplo 10. Faça um algoritmo que mostre a soma dos 10 primeiros números inteiros.

```
algoritmo "semnome"
var
  a: inteiro
inicio
  para a <-10 ate 1 passo -1 faça
    escreva(a)
  fimpara
fimalgoritmo
```

Exemplo 11. Faça um algoritmo que mostre os números de x ate z.

```
algoritmo "semnome"
var
  a , x, z: inteiro
inicio
  leia (x , z)
  para a <- x ate z faça
    escreva (a)
  fimpara
fimalgoritmo
```

Exemplo 11. Faça um programa para calcular N!. (**)

```
algoritmo "Fatorial"
var
    numero, fatorial, contador: inteiro

inicio
    fatorial <- 1
    leia (numero)
    Para contador de 1 ate numero faca
        fatorial <- fatorial * contador
    fimpara
    escreva(fatorial)
fimalgoritmo
```

Exemplo 12. Leia n números e mostre sua soma e media.

```
algoritmo "semnome"
var
    a, n : inteiro
    soma , numero: inteiro
    media: real
inicio
    leia (n)
    soma <- 0

    para a <- 1 ate n faca
        leia (numero)
        soma <- soma + numero
    fimpara

    media <- soma/numero
    escreva (soma)
    escreva (media)

fimalgoritmo
```

Exemplo 13. Leia n números e mostre o maior.

```
algoritmo "semnome"
var
    a, numero, maior, n : inteiro

inicio
    leia (n)

    se n > 0 entao
        leia (numero)
        maior <- numero

        para a <- 1 ate n-1 faca
            leia (numero)
            se numero > maior entao
                maior <- numero
            fimse
        fimpara

        escreva(maior)
    fimse
fimalgoritmo
```

Exercícios

- 1) Faça um algoritmo que leia 10 números e mostre a soma e a média dos que forem maiores que o valor 100.
- 2) Um contador precisa de um algoritmo em que ele indique o número de salários recebidos por um funcionário e o algoritmo pergunte cada um destes salários e ao final mostre o número de salários digitados, a soma deles e a média.
- 3) Faça os seguintes Exercícios Resolvidos do livro texto (Capítulo 4, a partir da página 83). Leia o enunciado e elabore uma solução própria. Use a solução apresentada apenas como referência. Procure seguir a ordem sugerida.
 - a) Exercício 14: soma dos números pares.
 - b) Exercício 9: funcionários temporários.
 - c) Exercício 25: números nos intervalos.
 - d) Exercício 10: série de Fibonacci.
 - e) Exercício 3: fatorial de N valores lidos.
- 4) Faça um programa para ler um valor X e um valor n. Após, calcule a seguinte expressão:

$$Y = \frac{(X+1)}{1!} + \frac{(X+2)}{2!} + \frac{(X+3)}{3!} + \dots + \frac{(X+N)}{N!}$$

- 5) Faça um programa para calcular um valor A elevado a um expoente B. Os valores A e B deverão ser lidos. Usar apenas soma. Não usar exponenciação.